

Projekt: Werkstückanschlag

Aufgabe: Bolzen Pos. 3


Programmierung: Heidenhain TNC 7

Autor:

Klasse:

Datum:

```
0 BEGIN PGM POS3_BOLZEN_FERTIG MM
1 BLK FORM 0.1 Z X-10 Y-16.5 Z-20
2 BLK FORM 0.2 X+10 Y+16.5 Z+0
3 ;
4 ;Werkzeug NC-Anbohrer D=12mm
5 ;Werkzeugaufruf
6 TOOL CALL "T6_NC_ANBO_D12" Z S3000
7 M3
8 ;
9 ;Positionieren
10 L Z+100 R0 FMAX
11 ;
12 ;Zyklusdefinition Bohren
13 CYCL DEF 200 BOHREN ~
    Q200=+2 ;SICHERHEITS-ABST. ~
    Q201=-4.5 ;TIEFE ~
    Q206=+150 ;VORSCHUB TIEFENZ. ~
    Q202=+1 ;ZUSTELL-TIEFE ~
    Q210=+0 ;VERWEILZEIT OBEN ~
    Q203=+0 ;KOOR. OBERFLAECHE ~
    Q204=+10 ;2. SICHERHEITS-ABST. ~
    Q211=+0 ;VERWEILZEIT UNTEN
14 ;
15 ;Positionieren und Zyklusaufruf
16 L X+0 Y+0 R0 FMAX M99 M8
17 ;
18 ;Freifahren
19 L Z+100 R0 FMAX M9
20 ;
21 ;Werkzeug Bohrer D=8,5mm
22 ;Werkzeugaufruf
23 TOOL CALL "T10_TL_D8_5" Z S1800
24 M3
25 ;
26 ;Positionieren
27 L Z+100 R0 FMAX
28 ;Zyklusdefinition Bohren
29 CYCL DEF 200 BOHREN ~
    Q200=+2 ;SICHERHEITS-ABST. ~
    Q201=-23 ;TIEFE ~
    Q206=+250 ;VORSCHUB TIEFENZ. ~
    Q202=+3 ;ZUSTELL-TIEFE ~
    Q210=+0 ;VERWEILZEIT OBEN ~
```



```

    Q203=+0      ;KOOR. OBERFLAECHE ~
    Q204=+10     ;2. SICHERHEITS-ABST. ~
    Q211=+0      ;VERWEILZEIT UNTEN
30 ;
31 ;Positionieren und Zyklusaufruf
32 L  X+0  Y+0 R0 FMAX M99 M8
33 ;
34 ;Freifahren
35 L  Z+100 R0 FMAX M9
36 ;
37 ;Werkzeug SFR VHM D=8
38 ;Werkzeugaufruf
39 TOOL CALL "T2_SF_D8" Z S6000
40 M3
41 ;
42 ;Positionieren
43 L  Z+100 R0 FMAX
44 ;
45 ;Zyklusdefinition Kreistasche
46 CYCL DEF 252 KREISTASCHE ~
    Q215=+0      ;BEARBEITUNGS-UMFANG ~
    Q223=+13.5   ;KREISDURCHMESSER ~
    Q368=+0.1    ;AUFMASS SEITE ~
    Q207=+500    ;VORSCHUB FRAESEN ~
    Q351=+1      ;FRAESART ~
    Q201=-10     ;TIEFE ~
    Q202=+3      ;ZUSTELL-TIEFE ~
    Q369=+0      ;AUFMASS TIEFE ~
    Q206=+250    ;VORSCHUB TIEFENZ. ~
    Q338=+0      ;ZUST. SCHLICHTEN ~
    Q200=+2      ;SICHERHEITS-ABST. ~
    Q203=+0      ;KOOR. OBERFLAECHE ~
    Q204=+10     ;2. SICHERHEITS-ABST. ~
    Q370=+1      ;BAHN-UEBERLAPPUNG ~
    Q366=+1      ;EINTAUCHEN ~
    Q385=+500    ;VORSCHUB SCHLICHTEN
47 ;
48 ;Positionieren und Zyklusaufruf
49 L  X+0  Y+0 R0 FMAX M99 M8
50 ;
51 ;Freifahren
52 L  Z+30 R0 FMAX
53 ;
54 ;Werkzeug wegbringen
55 TOOL CALL 0 Z
56 ;
57 ;Programmende
58 M30
59 END PGM POS3_BOLZEN_FERTIG MM

```